Newsgroup: it.comp.appl.access
Titolo: compressione unicode - è possibile senza ricorrere a create table?

Salve a tutti,
utilizzo per la creazione di tabelle via codice il seguente:
 Function creatable()
Dim dbs As Database
Dim tdfNew As TableDef
         Set dbs = CurrentDb()
         Set tdfNew = dbs.CreateTableDef("miatabella")
                    tdfNew.Fields.Append tdfNew.CreateField("campo1", dbText, 16)
                    tdfNew.Fields.Append tdfNew.CreateField("campo2", dbText, 2)
                    tdfNew.Fields.Append tdfNew.CreateField("campo3", dbText, 13)
                    tdfNew.Fields.Append tdfNew.CreateField("campo4", dbText, 1)
                    tdfNew.Fields.Append tdfNew.CreateField("data", dbDate)
                    tdfNew.Fields.Append tdfNew.CreateField("denominazione", dbText, 65)
               dbs.TableDefs.Append tdfNew
               dbs.Close
           Set dbs = Nothing
End Function
 Avrei pero' necessità di specificare che ai campi testo deve essere
applicata la compressione unicode
ho provato ad aggiungere qualcosa del tipo
tdfNew.fields ("campo1").Properties("Unicode Compression") = true
ma ovviamente mi manda a quel paese.
si può realizzare in qualche modo?
grazie a tutti per la collaborazione
Antonio
--------------------------------
Inviato via 
http://arianna.libero.it/usenet/

 si può realizzare in qualche modo?
È solo possibile con ADO (CREATE TABLE  con "WITH COMP")
o ADOX (
http://support.microsoft.com/?id=275252
).
Non con DAO.
-- 
HTH
Karl
*********
Access FAQ: 
www.donkarl.com/it
 

 ha scritto nel messaggio
news:193Z203Z232Z5Y...@usenet.libero.it...

Ciao Antonio,
la proprietà si chiama UnicodeCompression e siccome non viene creata quando
aggiungi i campi, non pui variarla, devi crearla in seguito.
esempio
----
Function creatable()
Dim dbs As Database
Dim tdfNew As TableDef
         Set dbs = CurrentDb()
         Set tdfNew = dbs.CreateTableDef("miatabella")
         tdfNew.Fields.Append tdfNew.CreateField("campo1", dbText, 16)
         tdfNew.Fields.Append tdfNew.CreateField("campo2", dbText, 2)
         tdfNew.Fields.Append tdfNew.CreateField("campo3", dbText, 13)
         tdfNew.Fields.Append tdfNew.CreateField("campo4", dbText, 1)
         tdfNew.Fields.Append tdfNew.CreateField("data", dbDate)
         tdfNew.Fields.Append tdfNew.CreateField("denominazione", dbText,
65)
         dbs.TableDefs.Append tdfNew
         'Aggiungo la proprietà UnicodeCompression su campo1
         dbs.TableDefs("miatabella").Fields("campo1").Properties.Append _
             dbs.CreateProperty("UnicodeCompression", dbBoolean, True)
         'Aggiungo la proprietà UnicodeCompression su campo2
         dbs.TableDefs("miatabella").Fields("campo2").Properties.Append _
             dbs.CreateProperty("UnicodeCompression", dbBoolean, True)
         'ecc ecc
         dbs.Close
         Set dbs = Nothing
End Function
----
Ciao
-- 
Giorgio Rancati
[Office Access MVP]

 Non con DAO.
Ok, solo perché io non l'ho mai fatto non vuol dire che non
è possibile con DAO - come mostrato da Giorgio.
-- 
;-)


 ha scritto:

Infinite grazie...e chiedo scusa per rispondere e ringraziare solo adesso,
ma non sono stato in ufficio.
ovviamente funziona tutto....grazie sempre a tutti per la disponibilità
